

ANALIZA ELEKTRONIČKIH SKLOPOVA POMOĆU RAČUNALA KORIŠTENJEM PROGRAMA SPICE

Uvod

Složenost analize elektroničkih sklopova uvjetovana je prvenstveno nelinearnošću karakteristika elektroničkih elemenata. Zbog toga se pri analitičkom proračunu sklopova uvode znatna pojednostavljenja. Tako se npr. u analizi analognih sklopova karakteristike elemenata lineariziraju. Pri tome treba voditi računa da je točnost tog pristupa ograničena samo na režim malog signala. Linearizacija osigurava relativno jednostavan proračun sklopa u području srednjih frekvencija. Analiza se međutim znatno komplicira ako se proračun sklopa proširi na područja niskih ili visokih frekvencija, naročito ako sklop sadrži veći broj stupnjeva. Pri analizi digitalnih sklopova problemi analitičkog proračuna u pravilu su još veći. U radu digitalnih sklopova elementi prolaze kroz različita područja rada čime je njihova nelinearnost izraženija nego kod analognih sklopova.

Zbog niza pojednostavljenja rezultati analitičkog proračuna predstavljaju prvu aproksimaciju rješenja, koje se u praksi provjerava mjerenjem na gotovom sklopu. To je uobičajeno kod diskretne realizacije sklopa. U tom slučaju moguće je mjeriti signal u bilo kojoj točki, te po potrebi zamijeniti pojedini element. U integriranoj izvedbi sklopa te mogućnosti ne postoje. Upravo stoga s početkom razvoja integriranih sklopova pojavila se potreba za točnijom analizom sklopova koja će se izvoditi prije njegove realizacije, a čiji će rezultati u što većoj mjeri odgovarati stvarnim karakteristikama sklopa. Točniju analizu omogućuje numerički proračun sklopa pomoću računala. Ovaj pristup nudi nove mogućnosti u odnosu na analitički proračun. Prije svega u opisu sklopa uključuju se složeniji, ali znatno realniji nelinearni modeli elektroničkih elemenata koji osiguravaju točnije rezultate analize. Prednosti numeričkog proračuna naročito su izražene pri vremenskoj ili frekvencijskoj analizi većih sklopova, čiji analitički proračun najčešće nije moguć.

Razvijen je veći broj programa specijaliziranih za numerički proračun elektroničkih sklopova pomoću računala. Daleko najpoznatiji i najčešće korišten program ove vrste je **SPICE**, čije je ime kratica punog engleskog naziva “**S**imulation **P**rogram with **I**ntegrated **C**ircuits **E**mphasis”. Iako naziv upućuje da je njegov razvoj bio prvenstveno motiviran analizom integriranih elektroničkih sklopova, SPICE se jednako uspješno primjenjuje za analizu elektroničkih sklopova u bilo kojoj tehnološkoj realizaciji.

SPICE je općeniti naziv za cijelu skupinu programa, koji se temelje na programu **SPICE2**. SPICE2 je dovršen 1975. godine na University of California, Berkeley, USA. Od tada se intenzivno razvija, te je postao industrijski standard za električku analizu sklopova. Svoju popularnost u velikoj mjeri zahvaljuje činjenici da se razvojem elektroničkih elemenata u program stalno ugrađuju novi i nadopunjeni modeli koji opisuju njihovo ponašanje. Osim toga u programu se često uvode i nove mogućnosti analize sklopova. Danas se SPICE, osim na sveučilištima, razvija i u komercijalnim tvrtkama. Komercijalne izvedbe programa najčešće su nadopunjene pomoćnim grafičkim alatima za unos električke sheme, te analizu rezultata, čime se znatno olakšava njihovo korištenje. Među najpoznatijim komercijalnim izvedbama su: **HSPICE** (Meta-Software) i **IG-SPICE** (A.B. Associates) za veća UNIX računala, te **PSpice** (MicroSim) i **IS-SPICE** (Intusoft) za osobna računala.

Nastavak teksta opisuje program SPICE u obimu koji je nužan za obavljanje laboratorijskih vježbi. Ukupne mogućnosti programa znatno su veće i mogu se naći u brojnoj literaturi [1-3].

Vrste analize

Kao sklopovski simulator opće namjene SPICE obavlja tri osnovne vrste analize:

- nelinearnu istosmjernu analizu,
- nelinearnu vremensku analizu,
- linearnu frekvencijsku analizu.

Nelinearnom istosmjernom analizom (**DC analizom**) proračunava se ponašanje sklopa uz primjenu istosmjernih napona i struja. Pri toj analizi induktiviteti su kratko spojeni, a kapaciteti odspojeni. Često analiza sklopa započinje istosmjernom analizom, čiji je rezultat statička radna točka ili radna točka karakteristike sklopa. Specijalni slučaj DC analize predstavlja proračun istosmjerne velikosignalne prijenosne karakteristike. Ovaj proračun daje skup istosmjernih točaka koje su rezultat postupne promjene vrijednosti neovisnog naponskog ili strujnog izvora. Isti postupak analize može se izvesti uz postupnu promjenu temperature sklopa (**temperaturna analiza**).

Pri nelinearnoj vremenskoj analizi (**tranzijentnoj analizi**) računaju se naponi i struje sklopa u zadanom vremenskom intervalu kao rezultat primjene vremenski promjenljivih ulaznih signala. Kao dodatak ove analize može se izvršiti **Fourierova analiza**, koja računa Fourierove koeficijente pojedinih napona ili struja sklopa u odnosu na osnovnu frekvenciju ulaznog signala.

Linearna frekvencijska analiza (**AC analiza**) je analiza malog signala. Pri ovoj se analizi sklop linearizira u statičkoj radnoj točki, a naponi i struje računaju se u ovisnosti o frekvenciji ulaznih sinusoidnih signala. Ispravnost primjene linearnog sklopa temelji se na pretpostavci da su ulazni signali malih amplituda. AC analiza može se nadopuniti **analizom šuma**.

Proračun utjecaja tolerancija elemenata na rad sklopa omogućuju dvije vrste analiza. Pri istosmjernoj **analizi osjetljivosti** promjena radne točke računa se u ovisnosti o malim promjenama vrijednosti pojedinih komponenata sklopa. **Monte Carlo analiza** izvodi niz DC, AC ili tranzijentnih analiza sklopa uz statističku raspodjelu vrijednosti pojedinih komponenata.

Ulazni podaci programa

Sklop predviđen za simulaciju programom SPICE opisuje se ulaznom datotekom. Ulazna datoteka je tekstualna (ASCII) i sastoji se od niza redaka. Rec i su opisni i kontrolni. Opisni rec i ili rec i elemenata sadrže pojedine elemente sklopa i jednoznačno opisuju njegovu topologiju. Kontrolni rec i definiraju parametre modela, tipove analize i način prikaza izlaznih rezultata.

Prvi redak u ulaznoj datoteci program prihvaća kao naslov i ispisuje ga u zaglavlju rezultata. Taj redak ne smije sadržavati podatke o sklopu, jer će ih program ignorirati. Zadnji redak u ulaznoj datoteci mora biti kontrolna naredba **.END**. Redoslijed ostalih redaka je proizvoljan.

Dozvoljena je upotreba komentarskih redaka, koji poboljšavaju čitljivost ulazne datoteke. Komentarski rec i započinju znakom *****.

Format zapisa u recima je slobodan. Pojedini se podaci (riječi) odvajaju s jednim ili više razmaka, sa zarezom ili s kombinacijom zareza i razmaka. Dozvoljena dužina retka je 80 znakova. Sadržaj retka može se nastaviti u slijedećem retku. Redak koji je nastavak prethodnog retka započinje znakom **+**. Dozvoljena je upotreba velikih i malih slova, koje program međusobno ne razlikuje.

U ulaznoj datoteci brojevi se mogu zapisivati na više načina; kao:

- cijeli brojevi (npr. 5),
- brojevi s pomičnom točkom (npr. 55.5),
- brojevi s cjelobrojn timer eksponentom (npr. 5E5, 5.5E-5),

- brojevi sa skalirajućim faktorom.

Skalirajući faktori prikazani su u Tablici 1.

Tablica 1 Skalirajući faktori

Oznaka	Naziv	Faktor množenja
T	tera	10^{+12}
G	giga	10^{+9}
Meg	mega	10^{+6}
K	kilo	10^{+3}
M	mili	10^{-3}
U	mikro	10^{-6}
N	nano	10^{-9}
P	piko	10^{-12}
F	femto	10^{-15}

Skalirajući faktori neposredno slijede broj kojemu su pridruženi. Slova neposredno iza broja koja nisu navedeni faktori, kao i slova neposredno iza faktora program ignorira. Ta slova mogu se koristiti radi bolje čitljivosti podataka (npr. $5V = 5$, $20mA = 20E-3$).

Pri zadavanju ulaznih podataka i interpretaciji izlaznih rezultata treba voditi računa da SPICE koristi MKS sustav jedinica.

Opis sklopa

Unos topologije sklopa u ulazne podatke programa započinje označavanjem svih čvorova u električkoj shemi. Čvorovi se najčešće označavaju pozitivnim cijelim brojevima, ali se mogu označiti i nizom alfanumeričkih znakova (npr. UCC, ul). Ukoliko se čvorovi označavaju pozitivnim cijelim brojevima, brojevi ne moraju biti numerirani u nizu. Sklop mora sadržavati referentni čvor (masu) označen s **0**.

Svaki od elemenata sklopa opisuje se jednim retkom elementa. SPICE razlikuje veći broj elemenata, koji se mogu podijeliti u četiri grupe: pasivni elementi, neovisni izvori, ovisni izvori i elektronički elementi. Rec i elemenata započinju s podatkom od maksimalno 8 alfanumeričkih znakova, među kojima ne smiju biti posebni znakovi (znak razdvajanja, znak jednakosti, zarez, točka, zagrada). Prvi znak, koji mora biti jedno od slova iz tablice 2, označuje tip elementa. Ostali znakovi predstavljaju ime elementa. Slijedeći podaci u retku elementa opisuju čvorove između kojih je element spojen, te njegova svojstva.

Pasivni elementi

Najčešće korišteni pasivni elementi su kondenzator (**C**), zavojnica (**L**) i otpornik (**R**). Opisuju se sa:

Cime	n1 n2 vrijednost
Lime	n1 n2 vrijednost
Rime	n1 n2 vrijednost

Iako kod ovih elemenata redoslijed čvorova najčešće nije bitan, SPICE interpretira prvi čvor **n1** kao pozitivan, a drugi čvor **n2** kao negativan. Pozitivna struja teče pri tome od prvog čvora **n1**, kroz element u drugi čvor **n2**. Vrijednosti elemenata su: kapacitet u F za kondenzator, induktivitet u H za zavojnicu, te otpor u Ω za otpornik. Primjeri redaka kojima se opisuju pasivni elementi su:

Tablica 2 Osnovni elementi

Prvo slovo	Element
C	kondenzator
D	dioda
E	naponski upravljani naponski izvor
F	strujno upravljani strujni izvor
G	naponski upravljani strujni izvor
H	strujno upravljani naponski izvor
I	neovisan strujni izvor
J	spojni FET
K	međuiduktivitet
L	zavojnica
M	MOSFET
Q	bipolarni tranzistor
R	otpornik
V	neovisan naponski izvor

Cvezni 5 0 10uF

L1 3 4 1mH

RE 5 0 1kohm

Neovisni izvori

U SPICE-analizi koriste se tri tipa neovisnih izvora: istosmjerni (DC), frekvencijski promjenjivi (AC), te različiti valni oblici vremenski promjenjivih izvora. Pri tome svaki od ovih tipova izvora može biti strujni (**I**) ili naponski (**V**). Opći oblik opisa neovisnih izvora je:

Iime **n+ n- DC vrijednost**
+ **AC amplituda faza specifikacija**
Vime **n+ n- DC vrijednost**
+ **AC amplituda faza specifikacija**

n+ je pozitivan, a **n-** negativan čvor izvora. Pozitivna struja teče od pozitivnog čvora **n+**, kroz izvor u negativni čvor **n-**.

Oznaku **DC** istosmjernog izvora nije nužno navesti. Iznos struje ili napona istosmjernog izvora zadaje se kao **vrijednost**. Frekvencijski promjenjivi izvor je sinusnog valnog oblika i označuje se s **AC**. Za taj se tip izvora obavezno zadaje **amplituda**. Ako se ne zada **faza** SPICE pretpostavlja da je njena vrijednost 0.

Parametrom **specifikacija** opisuje se vremenski promjenjivi izvor. SPICE prihvata slijedeće oblike vremenski promjenjivih izvora:


EXP (parametri)	eksponencijalna pobuda
PULSE (parametri)	impulsna pobuda
PWL (parametri)	pobuda linearna po odsječcima
SFFM (parametri)	frekvencijski modulirana pobuda
SIN (parametri)	sinusna pobuda.

Za laboratorijske vježbe interesantni su valni oblici impulsne pobude, pobude linearne po odsječcima i sinusne pobude.

Opći oblik **impulsne** pobude je:

PULSE (U1 U2 td tr tf tw T)

gdje su pojedini parametri opisani u valnom obliku na slici 1.




Slika 1. Valni oblik impulsne pobude

Pri opisu impulsnog valnog oblika moraju se zadati naponi **U1** i **U2**. Ukoliko se ne zadaju, parametar **td** poprima vrijednost 0, parametri **tr** i **tf** vrijednost **tkorak**, a parametri **tw** i **T** vrijednost **tkraj**, pri čemu su **tkorak** i **tkraj** parametri nelinearne vremenske analize.

Opći oblik pobude **linearne po odsječcima** je:

PWL (t1 U1 t2 U2 ... tn Un),

Napon se mijenja linearno po odsječcima od točke do točke. Svaka točka opisana je vremenom **t** i naponom **U** prema valnom obliku na slici 2.



Slika 2. Valni oblik pobude linearne po odsječcima


Opći oblik **sinusne** pobude je:

$$\text{SIN} (U_0 \ U_m \ f \ t_d \ d_f \ \text{phi}).$$

Valni oblik pobude opisan je izrazom

$$u = U_0 + U_m \sin \left\{ 2\pi \left[f(t - t_d) - \frac{\text{phi}}{360} \right] \right\} \exp[-d_f(t - t_d)]. \quad (1)$$

i prikazan je na slici 3.



Slika 3. Valni oblik sinusne pobude

Pri opisu sinusnog valnog oblika moraju se zadati naponi **U0** i **Um**. Ukoliko se ne zadaju, parametar **f** poprima vrijednost **1/tkraj**, a parametri **td**, **df** i **phi** vrijednost 0. Parametar **tkraj** je parametar nelinearne vremenske analize.

Svi navedeni vremenski promjenjivi izvori prikazani su kao naponski. Mogu se koristiti i strujni izvori istih valnih oblika.

Pojedini tipovi neovisnih izvora primjenjuju se u različitim analizama. Istosmjerni izvori koriste se u nelinearnoj istosmjernoj analizi, frekvencijski promjenjivi izvori u linearnoj frekvencijskoj analizi, a vremenski promjenjivi izvori u nelinearnoj vremenskoj analizi. Zbog toga isti redak ulaznih podataka može sadržavati različite tipove istog neovisnog izvora, koji se primjenjuju svaki za svoju analizu. Primjeri redaka kojima se opisuju neovisni izvori su:

```
VDC 5 0 10V
IAC 10 5 AC 1mA
VPLS 10 5 PULSE (-1V 1V 0 0.1us 0.1us 2us 10us)
IPWL 10 5 PWL (0 -10mA 1us 0A 10us 0A 10.1us 20mA)
VSIN 11 3 DC 5V AC 1 SIN(.002 .002 1.5MEG)
```

Linearno ovisni izvori

SPICE razlikuje 4 vrste linearno ovisnih izvora. To su naponski upravljani naponski izvor (**E**), strujno upravljani strujni izvor (**F**), naponski upravljani strujni izvor (**G**) i strujno upravljani naponski izvor (**H**). Opći oblik opisa tih izvora je:

```
Eime    n+ n- nc+ nc- vrijednost
Fime    n+ n- Vime vrijednost
```

Gime **n+ n- nc+ nc- vrijednost**
Hime **n+ n- Vime vrijednost**

Ovisni izvor spojen je između pozitivnog čvora **n+** i negativnog čvora **n-**. Kao i kod neovisnog izvora pozitivna struja izvora teče od pozitivnog čvora **n+**, kroz izvor u negativni čvor **n-**. S **nc+** i **nc-** označeni su upravljački čvorovi. Kod naponskih upravljanih izvora ovisni izvor upravljan je naponom između pozitivnog upravljačkog čvora **nc+** i negativnog upravljačkog čvora **nc-**. Kod strujno upravljanih izvora ovisni izvor upravljan je strujom koja teče između upravljačkih čvorova **nc+** i **nc-**. Budući da SPICE “ne zna” računati struju između dva čvora, između upravljačkih čvorova spaja se istosmjerni naponski neovisni izvor **Vime**

Vime **nc+ nc- vrijednost**

Parametar **vrijednost** određuje istosmjerni napon izvora **Vime**. Struja upravljačkog napona **Vime** teče od pozitivnog upravljačkog čvora **nc+**, kroz izvor u negativni upravljački čvor **nc-**.

Vrijednosti ovisnih izvora su: naponsko pojačanje za naponski upravljan naponski izvor, strujno pojačanje za strujno upravljan strujni izvor, strmina u A/V za naponski upravljan strujni izvor, te prijenosni otpor u V/A za strujno upravljan naponski izvor. Primjeri redaka kojima se opisuju linearno ovisni izvori su:

```
EAMP 1 2 10 11 10
G3 1 2 10 11 10m
```

Elektronički elementi

Zahvaljujući ugrađenim modelima SPICE omogućuje analizu sklopova s različitim elektroničkim elementima kao što su diode (**D**), bipolarni tranzistori (**Q**), spojni FET-ovi (**J**), MOSFET-ovi (**M**) i sl. Opći oblik opisa tih elemenata je:

Dime **na nk oznaka**
Qime **nc nb ne oznaka**
Jime **nd ng ns oznaka**
Mime **nd ng ns nb oznaka**

Opis je sličan kao i kod ostalih električkih elemenata. Nakon imena elementa navode se čvorovi između kojih je element spojen. Zbog nelinearnosti karakteristika redosljed čvorova je bitan i treba se unijeti točno prema gornjem opisu. Za diodu **na** je čvor anode, a **nk** čvor katode. Za bipolarni tranzistor **nc** je čvor kolektora, **nb** je čvor baze, a **ne** je čvor emitera. Kod FET-ova **nd** je čvor odvoda, **ng** je čvor upravljačke elektrode, a **ns** je čvor uvoda. Kod MOSFET-a dodatno se zadaje i čvor podloge **nb**.

Karakteristike elektroničkih elemenata opisuju se modelima, koji su definirani većim brojem parametara. U retku elementa ne navode se vrijednosti parametara, već se na tom mjestu daje samo ime modela **oznaka**. Kao **oznaka** može se upotrijebiti bilo koja kombinacija do 8 alfanumeričkih znakova, pri čemu prvi znak mora biti slovo. Na taj način definira se tip elementa, a njegovi parametri navode se u posebnom kontrolnom **.MODEL** retku. Ukoliko sklop sadrži više elemenata istog tipa, za svaki od tipova zadaje se samo jedan **.MODEL** redak. Primjeri redaka kojima se opisuju elektronički elementi su:

```
D5 14 0 DMOD
Q1 15 10 20 T14
```

Parametri elektroničkih elemenata

Parametri pojedinih tipova elektroničkih elemenata definiraju se posebnim retkom oblika

.MODEL oznaka tip (parametri)

Ime modela **oznaka** mora biti isto kao i u retku elemenata za koji se parametri definiraju. Parametrom **tip** definira tip pojedinog elektroničkog elementa. Pojedini tipovi za najčešće korištene elemente navedeni su tablici 3.

Tablica 3 Tipovi osnovnih elektroničkih elemenata

tip	tip elementa
D	dioda
NPN	npn bipolarni tranzistor
PNP	pnP bipolarni tranzistor
NJF	n-kanalni spojni FET
PJF	p-kanalni spojni FET
NMOS	n-kanalni MOSFET
PMOS	p-kanalni MOSFET

Iza imena modela i tipa navodi se lista **parametri**. Lista **parametri** može se navoditi unutar zagrade, ali se zagrada može i izostaviti. Za svaki od modela pojedinog elektroničkog elementa definirana je skupina parametara. Za sve parametre program SPICE sadrži ugrađene vrijednosti. U listi **parametri** nužno je navesti samo one parametre čije su vrijednosti različite od ugrađenih. Svaki od parametra navodi se u obliku **ime=vrijednost**. Redoslijed parametara pri tome nije važan. Primjer **.MODEL** retka je:

```
.MODEL T3858 npn IS=5.07fA BF=157 VAF=244V IKF=64.5mA ISE=314FA
+          ne=1.56 rb=3000HM rc=20HM Cje=12.3PF Vje=0.89V
+          mje=0.404 tf=813ps Cjc=7PF Vjc=0.6 mjc=.333
```

Isti tipovi elektroničkih elemenata mogu se koristiti u različitim sklopovima. U tom je slučaju jednostavnije upisati te parametre u formi **.MODEL** redaka u posebnu datoteku. Ako su parametri elektroničkog elementa koji se koristi u sklopu sadržani u takvoj datoteci pri opisu sklopa može se koristiti redak

.LIB datoteka

U parametru **datoteka** treba navesti ime datoteke koja sadrži parametre elektroničkih elemenata. Pri tome ime modela **oznaka** u datoteci mora biti isto kao i u retku elemenata za koji se parametri definiraju.

Kontrola izvođenja programa

Uz opis sklopa u ulaznu datoteku treba upisati i vrste analiza. U tu svrhu koriste se kontrolni reci ili naredbe za koje je karakteristično da započinju točkom “.”.

Nelinearna istosmjerna analiza

Naredbom

.OP

zahtijeva se proračun statičke radne točke. Kao rezultat ove naredbe SPICE će u izlaznoj datoteci ispisati napone svih čvorova, struje svih izvora i ukupnu disipaciju snage.

Za proračun istosmjerne prijenosne karakteristike sklopa koristi se naredba

.DC ulaz početak kraj korak

SPICE će proračunavati sklop uz postupnu promjenu istosmjerne vrijednosti neovisnog izvora **ulaz**. Promjena vrijednosti je linearna od **početak** do **kraj** za iznos **korak**. Ako se neovisni izvor **ulaz** zamijeni s **TEMP**, SPICE izvodi analizu sklopa uz postupnu promjenu temperature. Primjer ove vrste naredbe je:

```
.DC VUL -0.5V 0.5V 50mV
```

Nelinearna vremenska analiza

Naredbom

```
.TRAN tkorak tkraj
```

izvodi se proračun sklopa u ovisnosti o vremenu. Vrijeme se mijenja od $t=0$ do **tkraj**. Program sam podešava vremenski korak proračuna, a **tkorak** je vremenski interval koji se koristi za ispis ili crtanje rezultata analize. Prije prijelaza na vremensku analizu SPICE proračunava radnu točku u kojoj određuje početne vrijednosti. Primjer naredbe za nelinearnu vremensku analizu je:

```
.TRAN 1ns 100ns
```

U okviru nelinearne vremenske analize SPICE može proračunati Fourierove koeficijente napona ili struja. Zahtjev za Fourierovu analizu zadaje se s

```
.FOUR f izlazi
```

S **f** zadaje se osnovna frekvencija Fourierove analize, a izlazne varijable **izlazi** su istog oblika kao i u **.PRINT** naredbi. Minimalni vremenski interval za izvedbu ove analize **tkraj = 1/f**. Rezultati Fourierove analize dobivaju se u izlaznoj datoteci. Uz istosmjernu vrijednost, ispisuju se apsolutne i relativne amplitude osnovnog i prvih 9 slijedećih harmonika, faze tih komponenata, te ukupni faktor izobličenja. Primjer naredbe za Fourierovu analizu je:

```
.FOUR 10kHz V(5) I(R1)
```

Linearna frekvencijska analiza

Ova vrsta analiza pokreće se jednom od naredbi

```
.AC DEC nd fpočetak fkraj
```

```
.AC OCT no fpočetak fkraj
```

```
.AC LIN nu fpočetak fkraj
```

Nakon linearizacije sklopa u radnoj točki, SPICE izvodi analizu uz promjenu frekvencije neovisnih **AC** izvora. Frekvencija se mijenja od **fpočetak** do **fkraj**. Uz parametre **DEC** ili **OCT** promjena frekvencije je logaritamska s **nd** koraka po dekadi ili **no** koraka po oktavi. Parametrom **LIN** zahtjeva se linearna promjena frekvencije s ukupnim brojem koraka **nu**. Primjer naredbe za linearnu frekvencijsku analizu je:

```
.AC DEC 10 1kHz 100MHz
```

Oblik izlaznih rezultata

Prilikom analize sklopa SPICE generira izlaznu datoteku u koju se ispisuju rezultati. Ispis rezultata određuju kontrolne naredbe u ulaznoj datoteci. Tako npr. naredba **.OP** inicira ispis napona čvorova u radnoj točki, a naredba **.FOUR** ispis Fourierovih koeficijenata.

Naredbama tipa **.DC**, **.TRAN** i **.AC** zadaju se promjene nezavisnih veličina koje mogu biti istosmjerne vrijednosti napona ili struja, vrijeme ili frekvencija. Pri izvođenju tih analiza SPICE izračunava sve napone i struje sklopa u ovisnosti o promjeni nezavisnih veličina. Prikaz takvih rezultata može se dobiti njihovim ispisom ili crtanjem.

Ispis rezultata u izlaznoj datoteci izvodi se naredbama oblika

```
.PRINT DC izlazi
.PRINT TRAN izlazi
.PRINT AC izlazi
```

Parametrima **DC**, **TRAN** odnosno **AC** ispisuju se rezultati DC, tranzijentne ili AC analize. Parametrom **izlazi** definiraju se naponi i struje u sklopu.

Mogu se odabirati naponi između dva čvora u sklopu (npr. **V(2,3)**) ili padovi napona na elementima. Ako se navede jedan čvor (npr. **V(5)**) SPICE ispisuje napon tog čvora prema masi. Za pad napona na dvopolnim elementima dovoljno je navesti ime elementa (npr. **V(R1)**), dok je kod trolepnih i više-polnih elemenata potrebno navesti između kojih se priključaka traži pad napona (npr. **VBE(Q5)**). Mogu se definirati struje elemenata (npr. **I(R3)**), pri čemu je za trolepne i više-polne elemente potrebno navesti struju priključka (npr. **IB(Q2)**).

Rezultati **AC** analize su kompleksni brojevi. Odabir rezultata određuje se dodavanjem nastavka iza napona **V** ili struje **I**. Ukoliko je nastavak **M** ili ga nema ispisuje se amplituda (npr. **VM(1)**), s nastavkom **DB** odabire se amplituda u decibelima (npr. **VdB(5,1)**), s nastavkom **P** faza (npr. **VP(R3)**), s nastavkom **R** realni (npr. **IR(Vu1)**), a s nastavkom **I** imaginarni dio kompleksnog broja (npr. **VIC(C3)**). Primjeri naredbi ispisa su:

```
.PRINT DC V(3) V(2,3) V(R1) I(VIN) I(R2) IB(Q10) VBE(Q13)
.PRINT AC VM(2) VP(2) VM(3,4) VDB(5)
.PRINT TRAN V(3) I(R1)
```

Zamjenom naredbe **.PRINT** naredbom **.PLOT**, u izlaznoj datoteci dobiva se uz ispis i grafički prikaz rezultata. Kvaliteta takvog prikaza formiranog od alfanumeričkih znakova ograničena je. Komercijalne izvedbe SPICE-a posjeduju često pomoćne programe za grafički prikaz rezultata. U programu **PSpice** za tu se namjenu koristi program **Probe**. Za korištenje tih mogućnosti potrebno je u ulaznu datoteku uključiti naredbu

```
.PROBE
```

Primjenom ove naredbe **PSpice** pohranjuje sve rezultate DC, tranzijentne i AC analize u posebnu datoteku, koji se mogu naknadno iscrtavati pokretanjem programa **Probe**.

PRIMJENA PROGRAMA PSpice

Na laboratorijskim vježbama električka analiza sklopova obavljat će se demonstracijskom izvedbom programa **PSpice** tvrtke MicroSim predviđene za rad u Windows operacijskom sustavu. Uz glavni program **PSpice** ta izvedba sadrži i pomoćne programe koje se koriste za formiranje ulaznih podataka, grafički pregled rezultata analize, definiranje parametara nelinearnih elektroničkih elemenata i slično. U odnosu na punu profesionalnu izvedbu, demonstracijska izvedba ima neka ograničenja, koja nisu bitna za uspješno obavljanje laboratorijskih vježbi.

Ulazna datoteka

Električka shema sklopa može se unijeti grafički uz pomoć programa **Schematics**. U okviru tog programa definiraju se i zahtjevi na vrste analiza. Program **Schematics** automatski generira datoteku s prosječnom shemom i omogućuje pokretanje električke analize programom **PSpice**.

U okviru laboratorijskih vježbi neće se koristiti program **Schematics**. Ulazne datoteke za analizu sklopova formirat će se pomoću programa za unošenje teksta **Textedit**. Pri upisu podataka u ulaznu

datoteku treba izbjegavati primjenu tabulatora. Nakon upisa podataka sadržaj ulazne datoteke pohranjuje se odabirom **Save** ili **Save As** iz izbornika **File**. **Save** se odabire ako je već ranije definirano ime datoteke, a **Save As** u slučaju da se ime prvi puta definira prilikom pohranjivanja. Uz ime, ulazna datoteka ima nastavak **.cir** (npr. sklop1.cir).

Izvođenje analize

Električka analiza sklopa pokreće se programom **PSpice**. Odabirom **Open** iz izbornika **File** otvara se pomoćni prozor u kojem se izabire ulazna datoteka. Odabirom datoteke automatski započinje električka analiza sklopa koji je opisan u ulaznoj datoteci. Analiza se izvodi prema kontrolnim naredbama iz ulazne datoteke.

Završetkom analize program **PSpice** formira izlaznu datoteku, čije ime odgovara imenu ulazne datoteke, a nastavak je **.out** (npr. sklop1.out). Odabirom **Examine Output** iz izbornika **File** pokreće se program **Textedit** sa sadržajem izlazne datoteke. Izlazna datoteka može se pregledavati, a po potrebi i tiskati.

Ukoliko je ulazna datoteka sadržavala kontrolnu naredbu **.probe**, prilikom izvođenja program **PSpice** zapisuje rezultate analize u posebnu datoteku. Ime te datoteke isto je kao i ime ulazne datoteke, a nastavak je **.dat** (npr. sklop1.dat).

Grafički prikaz rezultata analize

Ukoliko je formirana datoteka s nastavkom **.dat**, odabirom **Run Probe** iz izbornika **File** u programu **PSpice** može se pokrenuti program **Probe** za grafički prikaz rezultata analize. Grafički se mogu prikazivati rezultati DC analize u ovisnosti o promjenljivom ulaznom signalu, rezultati tranzijentne analize u ovisnosti o vremenu, te rezultati AC analize u ovisnosti o frekvenciji.

Odabir vrste analize

Ako je programom **PSpice** načinjeno više vrsta analiza, pri pokretanju programa **Probe** otvara se pomoćni prozor **Analysis Type** u kojem treba odabrati vrstu analize za koju će se prikazivati rezultati (**AC**, **DC** ili **Transient**). Nakon pregleda rezultata jedne vrste analize, prijelaz na drugu ostvaruje se odabirom te vrste analize u izborniku **Plot**.

Odabir i označavanje krivulja

Otvaranjem programa **Probe** u glavnom se prozoru iscertava koordinatni sustav. Os apscisa sadrži nezavisnu varijablu s vrijednostima koje odgovaraju izvedenoj analizi. Odabir rezultata koji će se prikazati na osi ordinata izvodi se odabirom **Add** u izborniku **Trace**. Pri tome se otvara pomoćni prozor **Add Traces** u kojem su navedene sve veličine koje se mogu crtati. To su naponi svih čvorova i struje svih elemenata. Označavanjem jedne ili više njih, te pritiskom na **OK** odabrane se vrijednosti iscertavaju u koordinatni sustav. Svaka se krivulja iscertava svojom bojom i svakoj se pridružuje pripadna oznaka u toj boji (pravokutnik, romb ili sl.). Legenda za pojedine krivulje ispisuje se ispod koordinatnog sustava s lijeve strane. Ukoliko se krivulje želi označiti pripadnim oznakama treba odabrati **Options** u izborniku **Tools**, te u pomoćnom prozoru pod **Use Symbols** odabrati **Always**.

Osim pojedinačnih veličina, koje nudi u pomoćnom prozoru **Add Traces**, program **Probe** može crtati i neke druge rezultate. Tako se npr. mogu prikazati naponi između dva čvora (npr. **V(4,2)**), potencijali pojedinih priključaka tropskih i višepolnih elemenata (npr. **VB(Q1)**), te padovi napona između priključaka tih elemenata (npr. **VEB(Q1)**).

Is crtavanje složenijih rezultata

Rezultati AC analize kompleksni su brojevi te se, prilikom odabira, u pomoćnom prozoru **Add Traces** ispred odabrane veličine (napona ili struje) upisuje jedan od oblika iz tablice 4.

Tablica 4 Prikaz rezultata AC analize

oznaka	značenje
M	amplituda
dB	amplituda u decibelima
p	faza
r	realni dio
i	imaginarni dio

Ukoliko se ne upiše ništa is crtava se amplituda odabrane veličine.

Mogu se crtati i kombinirane krivulje, koje su rezultat primjene matematičkih relacija na izlazne rezultate iz pomoćnog prozora **Add Traces**. Najjednostavniji primjeri su zbroj, razlika, umnožak i kvocijent pojedinih veličina. To npr. omogućuje prikaz prijenosne funkcije, kao omjera izlazne i ulazne veličine, snage kao umnoška napona i struje i sl. Moguća je primjena i složenijih matematičkih funkcija. Neke od mogućih funkcija nad pojedinačnim rezultatom y prikazane su tablici 5.

Tablica 5 Osnovne funkcije nad pojedinačnim rezultatima

oznaka	funkcija
$\text{abs}(y)$	$ y $
$\text{sqrt}(z)$	$\sqrt{y}$
$\text{pwr}(y, z)$	y^z
$\text{exp}(y)$	e^y
$\text{log}(y)$	$\ln(y)$
$\text{log10}(y)$	$\log(y)$
$d(y)$	derivacija y po nezavisnoj varijabli
$s(y)$	integral y po nezavisnoj varijabli
$\text{rms}(y)$	srednja kvadratna vrijednost od y po nezavisnoj varijabli

Koordinatne osi i koordinatni sustavi

Prema vrijednostima rezultata program **Probe** automatski određuje intervale na koordinatnim osima. Intervali se mogu mijenjati u pomoćnim prozorima koji se otvaraju odabirom **X Axis Settings** ili **Y Axis Settings** u izborniku **Plot**. Mogu se odabrati i načini prikaza osi (**Linear** i **Log**), a u pomoćnom prozoru koji se otvara s **X Axis Settings** može se izabrati i nova varijabla na osi apscisa. Na taj se način može npr. crtati Nyquistov dijagram kao $\text{iy} = f(\text{ry})$.

Odabirom **Add Plot** iz izbornika **Plot** otvara se novi koordinatni sustav. U svakom koordinatnom sustavu mogu se prikazati različite veličine u ovisnosti o istoj nezavisnoj varijabli. Taj pristup pogodan je za prikaz amplitudne i fazne frekvencijske karakteristike.


Korištenje pokazivača

Odabirom **Cursor** u izborniku **Tools** aktiviraju se dva pokazivača u sjecištu horizontalnih i vertikalnih pravaca. Istovremeno se otvara pomoćni prozor **Probe Cursor** s tri retka **A1**, **A2** i **diff**. Recima **A1** i **A2** pokazuju koordinate točaka koje odgovaraju trenutnim položajima pokazivača, a u retku **diff** prikazane su razlike njihovih koordinata. Pokazivači se koriste za očitavanje točaka na krivuljama.

U početku je pokretan samo prvi pokazivač koji se pomiče po prvoj krivulji. Oznaka te krivulje je uokvirena u legendi. Pokazivač se pomiče ili uz pomoć miša (pomaci lijevo-desno uz istovremenu pritisnutu lijevu tipku) ili pomoću strelica $\leftarrow$ i $\rightarrow$ na tipkovnici. Prebacivanje prvog pokazivača na drugu krivulju obavlja se aktiviranjem simbola te krivulje mišem uz pritisak lijeve tipke. Aktiviranjem mišem uz pritisak desne tipke odabire se drugi pokazivač. Njegovo pomicanje obavlja se na isti način kao i pomicanje prvog pokazivača, ali uz primjenu desne tipke miša ili kombinaciju **Shift** $\leftarrow$ i **Shift** $\rightarrow$ na tipkovnici.

PRIMJER ANALIZE SKLOPA POMOĆU PROGRAMA PSpice

Kao primjer sklopa za analiza pomoću programa **PSpice** odabran je spoj diferencijskog pojačala prema slici 4. Na slici su označeni svi elementi i upisani svi čvorovi. Radi preglednosti na slici nisu naznačene vrijednosti pojedinih elemenata. Prema slici 4 napisana je ulazna datoteka za SPICE-analizu sklopa.



Slika 4. Primjer sklopa za analizu pomoću programa PSpice

Ulazna datoteka za SPICE-analizu.

Diferencijsko pojačalo

*

```
VCC 13 0 12V
VEE 14 0 -6V
Vul 1 0 ac 1mV sin 0 0.2V 1KHz
```

*

```
RB1 1 2 .5kohm
RB2 10 0 .5kohm
RC3 13 3 10kohm
RC4 13 8 10kohm
RE3 5 6 50ohm
RE4 7 6 50ohm
RE5 12 14 1.3kohm
R1 11 0 2.9kohm
R2 11 14 1.3kohm
```

*

```
Q1 3 2 4 Q2N2222
Q2 8 10 9 Q2N2222
Q3 3 4 5 Q2N2222
Q4 8 9 7 Q2N2222
Q5 6 11 12 Q2N2222
```

*

.lib eval.lib

*

```
.op
.DC Vul -0.25V 0.25V 5mV
.ac dec 10 1Hz 10MegHz
.tran 10us 1ms
.four 1kHz V(8)
```

*

.probe

*

.end

Prvi dio ulazne datoteke sadrži opis svih elemenata sklopa. U recima elemenata upisane su i njihove vrijednosti. Budući da će se izvoditi više vrsta analiza, naponskom izvoru ulaznog signala **Vu_L** pridružena su dva tipa. Za AC analizu previđen je frekvencijski promjenjivi izvor **AC** s amplitudom **1mV**, a za tranzijentnu analizu vremenski promjenjivi izvor sinusnog valnog oblika **SIN** amplitude **0.2V** i frekvencije **1kHz**. Svi tranzistori su istog tipa (oznaka **Q2N2222**), a njihovi parametri definirani su u datoteci **eval.lib**.

Prvi zahtjev za analizu sklopa, postavljen naredbom **.op**, je izračunavanje statičke radne točke. Kao rezultat **PSpice** ispisuje u izlaznoj datoteci napone svih čvorova, struje svih izvora, te ukupnu istosmjernu disipaciju sklopa:

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(1)	0.0000	(2)	-20.99E-06	(3)	7.3814	(4)	-.4933
(5)	-1.1164	(6)	-1.1395	(7)	-1.1164	(8)	7.3814
(9)	-.4933	(10)	-20.99E-06	(11)	-4.1483	(12)	-4.7912
(13)	12.0000	(14)	-6.0000				

VOLTAGE SOURCE CURRENTS
NAME CURRENT

```
VCC -9.237E-04
VEE 2.354E-03
Vul -4.198E-08
```


TOTAL POWER DISSIPATION 2.52E-02 WATTS

Osim toga u izlaznoj datoteci sadržani su za svaki tranzistor istosmjerne struje i naponi, te parametri hibridnog nadomjesnog π -sklopa lineariziranog u statičkoj radnoj točki:

**** BIPOLAR JUNCTION TRANSISTORS


NAME	Q1	Q2	Q3	Q4	Q5
MODEL	Q2N2222	Q2N2222	Q2N2222	Q2N2222	Q2N2222
IB	4.20E-08	4.20E-08	3.07E-06	3.07E-06	6.08E-06
IC	3.03E-06	3.03E-06	4.59E-04	4.59E-04	9.24E-04
VBE	4.93E-01	4.93E-01	6.23E-01	6.23E-01	6.43E-01
VBC	-7.38E+00	-7.38E+00	-7.87E+00	-7.87E+00	-3.01E+00
VCE	7.87E+00	7.87E+00	8.50E+00	8.50E+00	3.65E+00
BETADC	7.22E+01	7.22E+01	1.49E+02	1.49E+02	1.52E+02
GM	1.17E-04	1.17E-04	1.77E-02	1.77E-02	3.56E-02
RPI	7.46E+05	7.46E+05	9.47E+03	9.47E+03	4.73E+03
RX	1.00E+01	1.00E+01	1.00E+01	1.00E+01	1.00E+01
RO	2.69E+07	2.69E+07	1.79E+05	1.79E+05	8.34E+04
CBE	3.20E-11	3.20E-11	4.30E-11	4.30E-11	5.09E-11
CBC	3.24E-12	3.24E-12	3.17E-12	3.17E-12	4.21E-12
CBX	0.00E+00	0.00E+00	0.00E+00	0.00E+00	0.00E+00
CJS	0.00E+00	0.00E+00	0.00E+00	0.00E+00	0.00E+00
BETAAC	8.74E+01	8.74E+01	1.68E+02	1.68E+02	1.68E+02
FT	5.28E+05	5.28E+05	6.11E+07	6.11E+07	1.03E+08

Slijedi DC analiza kojom se određuje istosmjerna prijenosna karakteristika sklopa. U zahtjevu za tom analizom mijenja se istosmjerna vrijednost ulaznog napona **Vu1** od **-0.25V** do **0.25V** s korakom **5mV**. Rezultati se mogu pregledati pomoću programa **Probe**. Na slici 5 prikazana je promjena izlaznog napona **V(8)**.




Slika 5. Istosmjerna prijenosna karakteristika

Pri izvođenju AC analize sklopa frekvencija ulaznog napona **Vu1** mijenja se logaritamski od **1Hz** do **10MegHz** s **10** koraka po dekadi. Amplitudna (**dB(V(8)/V(1))**) i fazna (**p(V(8)/V(1))**) frekvencijska karakteristika naponskog pojačanja iscrtana programom **Probe** dana je na slici 6.



Slika 6. Amplitudna i fazna frekventijska karakteristika naponskog pojačanja

Za analizu izobličenja treba koristiti nelinearnu vremensku analizu. Ta je analiza pokrenuta naredbom **.tran** u trajanju do **1ms**. Izvor signala pri tome je ulazni napon je **Vu1** sinusnog valnog oblika amplitude **0.2V** i frekvencije **1kHz**. Programom **Probe** iscertan je izlazni napon **V(8)** i dobiven je valni oblik prema slici 7. Može se uočiti izobličenje koje je posljedica prevelikog ulaznog signala.



Slika 7. Valni oblik izlaznog napona

Kvantitativnu analizu izobličenja omogućuje naredba **.four** kojom se izvodi Fourierova analiza. Odabrana je analiza izlaznog napona **V(8)** s osnovnom frekvencijom **1kHz**. Rezultati analize ispisani su u izlaznoj datoteci. Istosmjerna komponenta može se usporediti s vrijednošću napona iz statičke analize. Uz to dane su apsolutne i relativne amplitude i faze prvih 9 harmonika, te ukupni faktor izobličenja.:

FOURIER COMPONENTS OF TRANSIENT RESPONSE V(8)

DC COMPONENT = 7.382093E+00

HARMONIC NO	FREQUENCY (HZ)	FOURIER COMPONENT	NORMALIZED COMPONENT	PHASE (DEG)	NORMALIZED PHASE (DEG)
1	1.000E+03	4.608E+00	1.000E+00	-6.556E-01	0.000E+00
2	2.000E+03	1.520E-03	3.299E-04	7.371E+01	7.436E+01
3	3.000E+03	4.257E-01	9.238E-02	-1.707E+00	-1.051E+00
4	4.000E+03	1.499E-03	3.253E-04	8.860E+01	8.925E+01
5	5.000E+03	2.774E-02	6.021E-03	5.927E-01	1.248E+00
6	6.000E+03	1.486E-03	3.225E-04	8.694E+01	8.760E+01
7	7.000E+03	1.924E-03	4.175E-04	1.189E+02	1.196E+02
8	8.000E+03	1.504E-03	3.264E-04	8.680E+01	8.746E+01
9	9.000E+03	1.543E-03	3.348E-04	1.028E+02	1.035E+02

TOTAL HARMONIC DISTORTION = 9.258166E+00 PERCENT

Literatura

- [1] G.W. Roberts, A.S. Sedra, *SPICE*, Oxford University Press, New York 1997
- [2] P.W. Tuinenga, *SPICE - A Guide to Circuit Simulation and Analysis Using PSpice*, Prentice Hall, New Jersey, 1988.
- [3] *MicroSim Pspice A/D*, Reference Manual, MicroSim Corporation, Irvine, California, 1996
- [4] P.Biljanović, *Poluvodički elektronički elementi*, Školska knjiga, Zagreb, 1996
- [5] P. Antognetti, G. Massobrio, *Semiconductor Device Modeling with SPICE*, McGraw-Hill, New York, 1988